

CONJOINT

```
CONJOINT [PLAN={* }]  
          {file}  
  
[/DATA={* }]  
      {file}  
  
/{SEQUENCE}=varlist  
  {RANK }  
  {SCORE }  
  
[/SUBJECT=variable]  
  
[/FACTORS=varlist['labels']] ([[DISCRETE[{MORE}]]]  
                                {LESS }]  
                                {LINEAR[{MORE}]}  
                                {LESS }]  
                                {IDEAL }  
                                {ANTIIDEAL }]  
    [values['labels']]])  
    varlist...  
  
[/PRINT={ALL** } [SUMMARYONLY]]  
        {ANALYSIS }  
        {SIMULATION }  
        {NONE }  
  
[/UTILITY=file]  
  
[/PLOT={{SUMMARY} [SUBJECT] [ALL]]]  
      {{NONE**}]
```

**Default if subcommand or keyword is omitted.

Example:

```
CONJOINT PLAN='CARPLAN.SAV'  
/FACTORS=SPEED (LINEAR MORE) WARRANTY (DISCRETE MORE)  
PRICE (LINEAR LESS) SEATS  
/SUBJECT=SUBJ /RANK=RANK1 TO RANK15 /UTILITY='UTIL.SAV'.
```

Overview

CONJOINT analyzes score or rank data from full-concept conjoint studies. A plan file generated by ORTHOPLAN or entered by the user describes the set of full concepts scored or ranked in terms of preference. A variety of continuous and discrete models is available to estimate utilities for each individual subject and for the group. Simulation estimates for concepts not rated can also be computed.

Options

Data Input. You can analyze data recorded as rankings of an ordered set of profiles, or cards, as the profile numbers arranged in rank order, or as preference scores of an ordered set of profiles.

Model Specification. You can specify how each factor is expected to be related to the scores or ranks.

Display Output. The output can include the analysis of the experimental data, results of simulation data, or both.

Writing an External File. An SPSS data file containing utility estimates and associated statistics for each subject can be written for use in further analyses or graphs.

Basic Specification

- The basic specification is CONJOINT, a PLAN or DATA subcommand, and a SEQUENCE, RANK, or SCORE subcommand to describe the type of data.
- CONJOINT requires two files: a plan file and a data file. If only the PLAN subcommand or the DATA subcommand, but not both, is specified, CONJOINT will read the file specified on the PLAN or DATA subcommand and use the working data file as the other file.
- By default, estimates are computed using the DISCRETE model for all variables in the plan file (except those named *STATUS_* and *CARD_*). Output includes Kendall's tau and Pearson's product-moment correlation coefficients measuring the relationship between predicted and actual scores. Significance levels for one-tailed tests are displayed.

Subcommand Order

- Subcommands can appear in any order.

Syntax Rules

- Multiple FACTORS subcommands are all executed. For all other subcommands, only the last occurrence is executed.

Operations

- Both the plan and data files can be external SPSS data files. In this case, CONJOINT can be used before a working data file is defined.
- The variable *STATUS_* in the plan file must equal 0 for experimental profiles, 1 for holdout profiles, and 2 for simulation profiles. Holdout profiles are judged by the subjects but are not used when CONJOINT estimates utilities. Instead, they are used as a check on the validity of the estimated utilities. Simulation profiles are factor-level combinations that are not rated by the subjects but are estimated by CONJOINT based on the ratings of

the experimental profiles. If there is no *STATUS_* variable, all profiles in the plan file are assumed to be experimental profiles.

- All variables in the plan file except *STATUS_* and *CARD_* are used by CONJOINT as factors.
- In addition to the estimates for each individual subject, average estimates for each split-file group identified in the data file are computed. The plan file cannot have a split-file structure.
- Factors are tested for orthogonality by CONJOINT. If all of the factors are not orthogonal, a matrix of Cramér's *V* statistics is displayed to describe the nonorthogonality.
- When SEQUENCE or RANK data are used, CONJOINT internally reverses the ranking scale so that the coefficients computed are positive.
- The plan file cannot be sorted or modified in any way once the data are collected, since the sequence of profiles in the plan file must match the sequence of values in the data file in a one-to-one correspondence. (CONJOINT uses the order of profiles as they appear in the plan file, not the value of *CARD_*, to determine profile order.) If RANK or SCORE is the data-recording method, the first response from the first subject in the data file is the rank or score of the first profile in the plan file. If SEQUENCE is the data-recording method, the first response from the first subject in the data file is the profile number (determined by the order of profiles in the plan file) of the most preferred profile.

Limitations

- Factors must be numeric.
- The plan file cannot contain missing values or case weights. In the working data file, profiles with missing values on the SUBJECT variable are grouped together and averaged at the end. If any preference data (the ranks, scores, or profile numbers) are missing, that subject is skipped.
- Factors must have at least two levels. The maximum number of levels for each factor is 99.

Example

```
CONJOINT PLAN='CARPLAN.SAV'
/FACTORS=SPEED (LINEAR MORE) WARRANTY (DISCRETE MORE)
PRICE (LINEAR LESS) SEATS
/SUBJECT=SUBJ /RANK=RANK1 TO RANK15 /UTILITY='UTIL.SAV'.
```

- The PLAN subcommand specifies the SPSS data file *CARPLAN.SAV* as the plan file containing the full-concept profiles. Since there is no DATA subcommand, the working data file is assumed to contain the subjects' rankings of these profiles.
- The FACTORS subcommand specifies the ways in which the factors are expected to be related to the rankings. For example, speed is expected to be linearly related to the rankings, so that cars with higher speeds will receive lower (more-preferred) rankings.
- The SUBJECT subcommand specifies the variable *SUBJ* in the working data file as an identification variable. All consecutive cases with the same value on this variable are combined to estimate utilities.

- The RANK subcommand specifies that each data point is a ranking of a specific profile and identifies the variables in the working data file that contain these rankings.
- UTILITY writes out an external data file named *UTIL.SAV* containing the utility estimates and associated statistics for each subject.

PLAN Subcommand

PLAN identifies the file containing the full-concept profiles.

- PLAN is followed by the name of an external SPSS data file containing the plan or an asterisk to indicate the working data file.
- If the PLAN subcommand is omitted, the working data file is assumed by default. However, you must specify at least one external SPSS data file on a PLAN or a DATA subcommand. The working data file cannot be specified as both the plan and the data file.
- The file is specified in the usual manner for your operating system.
- The plan file is a specially prepared file generated by ORTHOPLAN or entered by the user. The plan file can contain the variables *CARD_* and *STATUS_*, and it must contain the factors of the conjoint study. The value of *CARD_* is a profile identification number. The value of *STATUS_* is 0, 1, or 2, depending on whether the profile is an experimental profile (0), a holdout profile (1), or a simulation profile (2).
- The sequence of the profiles in the plan file must match the sequence of values in the data file (see “Operations” on p. 2).
- Any simulation profiles (*STATUS_=2*) must follow experimental and holdout profiles in the plan file.
- All variables in the plan file except *CARD_* and *STATUS_* are used as factors by CONJOINT.

Example

```
DATA LIST FREE /CARD_ WARRANTY SEATS PRICE SPEED STATUS_.
BEGIN DATA
1 1 4 14000 130 2
2 1 4 14000 100 2
3 3 4 14000 130 2
4 3 4 14000 100 2
END DATA.
ADD FILES FILE='CARPLAN.SAV' /FILE=*.
CONJOINT PLAN=* /DATA='DATA.SAV'
  /FACTORS=PRICE (ANTIIDEAL) SPEED (LINEAR) WARRANTY (DISCRETE MORE)
  /SUBJECT=SUBJ /RANK=RANK1 TO RANK15 /PRINT=SIMULATION.
```

- DATA LIST defines six variables—a *CARD_* identification variable, four factors, and a *STATUS_* variable.
- The data between BEGIN DATA and END DATA are four simulation profiles. Each one contains a *CARD_* identification number and the specific combination of factor levels of interest.
- The variable *STATUS_* is equal to 2 for all cases (profiles). CONJOINT interprets profiles with *STATUS_* equal to 2 as simulation profiles.

- The ADD FILES command joins an old plan file, *CARPLAN.SAV*, with the working data file. Note that the working data file is indicated last on the ADD FILES command so that the simulation profiles are appended to the end of *CARPLAN.SAV*.
- The PLAN subcommand on CONJOINT defines the new working data file as the plan file. The DATA subcommand specifies a data file from a previous CONJOINT analysis.

DATA Subcommand

DATA identifies the file containing the subjects' preference scores or rankings.

- DATA is followed by the name of an external SPSS data file containing the data or an asterisk to indicate the current working data file.
- If the DATA subcommand is omitted, the working data file is assumed by default. However, you must specify at least one external SPSS data file on a DATA or a PLAN subcommand. The working data file cannot be specified as both the plan and the data file.
- The file is specified in the usual manner for your operating system.
- One variable in the data file can be a subject identification variable. All other variables are the subject responses and are equal in number to the number of experimental and hold-out profiles in the plan file.
- The subject responses can be in the form of ranks assigned to an ordered sequence of profiles, scores assigned to an ordered sequence of profiles, or profile numbers in preference order from most to least liked.
- Tied ranks or scores are allowed. CONJOINT issues a warning if tied ranks are present and then proceeds with the analysis. Data recorded in SEQUENCE format, however, cannot have ties, since each profile number must be unique.

Example

```

DATA LIST FREE /SUBJ RANK1 TO RANK15.
BEGIN DATA
01 3 7 6 1 2 4 9 12 15 13 14 5 8 10 11
02 7 3 4 9 6 15 10 13 5 11 1 8 4 2 12
03 12 13 5 1 14 8 11 2 7 6 3 4 15 9 10
04 3 6 7 4 2 1 9 12 15 11 14 5 8 10 13
05 9 3 4 7 6 10 15 13 5 12 1 8 4 2 11
50 12 13 8 1 14 5 11 6 7 2 3 4 15 10 9
END DATA.
SAVE OUTFILE='RANKINGS.SAV'.
DATA LIST FREE /CARD_ WARRANTY SEATS PRICE SPEED.
BEGIN DATA
 1 1 4 14000 130
 2 1 4 14000 100
 3 3 4 14000 130
 4 3 4 14000 100
 5 5 2 10000 130
 6 1 4 10000 070
 7 3 4 10000 070
 8 5 2 10000 100
 9 1 4 07000 130
10 1 4 07000 100
11 5 2 07000 070
12 5 4 07000 070
13 1 4 07000 070
14 5 2 10000 070
15 5 2 14000 130
END DATA.
CONJOINT PLAN=* /DATA='RANKINGS.SAV'
  /FACTORS=PRICE (ANTIIDEAL) SPEED (LINEAR)
  WARRANTY (DISCRETE MORE)
  /SUBJECT=SUBJ /RANK=RANK1 TO RANK15.

```

- The first set of DATA LIST and BEGIN–END DATA commands creates a data file containing the rankings. This file is saved in the external file *RANKINGS.SAV*.
- The second set of DATA LIST and BEGIN–END DATA commands defines the plan file as the working data file.
- The CONJOINT command uses the working data file as the plan file and *RANKINGS.SAV* as the data file.

SEQUENCE, RANK, or SCORE Subcommand

The SEQUENCE, RANK, or SCORE subcommand is specified to indicate the way in which the preference data were recorded.

SEQUENCE *Each data point in the data file is a profile number, starting with the most-preferred profile and ending with the least-preferred profile. This is how the data are recorded if the subject is asked to order the deck of profiles from most to least preferred. The researcher records which profile number was first, which profile number was second, and so on.*

RANK *Each data point is a ranking, starting with the ranking of profile 1, then the ranking of profile 2, and so on. This is how the data are recorded if the*

subject is asked to assign a rank to each profile, ranging from 1 to n , where n is the number of profiles. A lower rank implies greater preference.

SCORE *Each data point is a preference score assigned to the profiles, starting with the score of profile 1, then the score of profile 2, and so on.* These types of data might be generated, for example, by asking subjects to use a Likert scale to assign a score to each profile or by asking subjects to assign a number from 1 to 100 to show how much they like the profile. A higher score implies greater preference.

- You must specify one, and only one, of these three subcommands.
- After each subcommand, the names of the variables containing the preference data (the profile numbers, ranks, or scores) are listed. There must be as many variable names listed as there are experimental and holdout profiles in the plan file.

Example

```
CONJOINT PLAN=* /DATA='DATA.SAV'
/FACTORS=PRICE (ANTIIDEAL) SPEED (LINEAR) WARRANTY (DISCRETE MORE)
/SUBJECT=SUBJ
/RANK=RANK1 TO RANK15.
```

- The RANK subcommand indicates that the data are rankings of an ordered sequence of profiles. The first data point after *SUBJ* is variable *RANK1*, which is the ranking given by subject 1 to profile 1.
- There are 15 profiles in the plan file, so there must be 15 variables listed on the RANK subcommand.
- The example uses the TO keyword to refer to the 15 rank variables.

SUBJECT Subcommand

SUBJECT specifies an identification variable. All consecutive cases having the same value on this variable are combined to estimate the utilities.

- If SUBJECT is not specified, all data are assumed to come from one subject and only a group summary is displayed.
- SUBJECT is followed by the name of a variable in the working data file.
- If the same SUBJECT value appears later in the data file, it is treated as a different subject.

FACTORS Subcommand

FACTORS specifies the way in which each factor is expected to be related to the rankings or scores.

- If FACTORS is not specified, the DISCRETE model is assumed for all factors.
- All variables in the plan file except *CARD_* and *STATUS_* are used as factors, even if they are not specified on FACTORS.

- FACTORS is followed by a variable list and a model specification in parentheses that describes the expected relationship between scores or ranks and factor levels for that variable list.
- The model specification consists of a model name and, for the DISCRETE and LINEAR models, an optional MORE or LESS keyword to indicate the direction of the expected relationship. Values and value labels can also be specified.
- MORE and LESS keywords will *not* affect estimates of utilities. They are used simply to identify subjects whose estimates do not match the expected direction.

The four available models are:

DISCRETE *No assumption.* The factor levels are categorical and no assumption is made about the relationship between the factor and the scores or ranks. This is the default. Specify keyword MORE after DISCRETE to indicate that higher levels of a factor are expected to be more preferred. Specify keyword LESS after DISCRETE to indicate that lower levels of a factor are expected to be more preferred.

LINEAR *Linear relationship.* The scores or ranks are expected to be linearly related to the factor. Specify keyword MORE after LINEAR to indicate that higher levels of a factor are expected to be more preferred. Specify keyword LESS after LINEAR to indicate that lower levels of a factor are expected to be more preferred.

IDEAL *Quadratic relationship, decreasing preference.* A quadratic relationship is expected between the scores or ranks and the factor. It is assumed that there is an ideal level for the factor, and distance from this ideal point, in either direction, is associated with decreasing preference. Factors described with this model should have at least three levels.

ANTIIDEAL *Quadratic relationship, increasing preference.* A quadratic relationship is expected between the scores or ranks and the factor. It is assumed that there is a worst level for the factor, and distance from this point, in either direction, is associated with increasing preference. Factors described with this model should have at least three levels.

- The DISCRETE model is assumed for those variables not listed on the FACTORS subcommand.
- When a MORE or LESS keyword is used with DISCRETE or LINEAR, a reversal is noted when the expected direction does not occur.
- Both IDEAL and ANTIIDEAL create a quadratic function for the factor. The only difference is whether preference increases or decreases with distance from the point. The estimated utilities are the same for these two models. A reversal is noted when the expected model (IDEAL or ANTIIDEAL) does not occur.
- The optional value and value label lists allow you to recode data and/or replace value labels. The new values, in the order in which they appear on the value list, replace existing values starting with the smallest existing value. If a new value is not specified for an existing value, the value remains unchanged.

- New value labels are specified in apostrophes or quotation marks. New values without new labels retain existing labels; new value labels without new values are assigned to values in the order in which they appear, starting with the smallest existing value.
- A table is displayed for each factor that is recoded, showing the original and recoded values and the value labels.
- If the factor levels are coded in discrete categories (for example, 1, 2, 3), these are the values used by CONJOINT in computations, even if the value labels contain the actual values (for example, 80, 100, 130). Value labels are never used in computations. You can recode the values as described above to change the coded values to the real values. Recoding does not affect DISCRETE factors but does change the coefficients of LINEAR, IDEAL, and ANTIIDEAL factors.
- In the output, variables are described in the following order:
 1. All DISCRETE variables in the order in which they appear on the FACTORS subcommand.
 2. All LINEAR variables in the order in which they appear on the FACTORS subcommand.
 3. All IDEAL and ANTIIDEAL factors in the order in which they appear on the FACTORS subcommand.

Example

```
CONJOINT DATA='DATA.SAV'
  /FACTORS=PRICE (LINEAR LESS) SPEED (IDEAL 70 100 130)
  WARRANTY (DISCRETE MORE)
  /RANK=RANK1 TO RANK15.
```

- The FACTORS subcommand specifies the expected relationships. A linear relationship is expected between price and rankings, so that the higher the price, the lower the preference (higher ranks). A quadratic relationship is expected between speed levels and rankings, and longer warranties are expected to be associated with greater preference (lower ranks).
- The *SPEED* factor has a new value list. If the existing values were 1, 2, and 3, 70 replaces 1, 100 replaces 2, and 130 replaces 3.
- Any variable in the plan file (except *CARD_* and *STATUS_*) not listed on the FACTORS subcommand uses the DISCRETE model.

PRINT Subcommand

PRINT controls whether your output includes the analysis of the experimental data, the results of the simulation data, both, or none.

The following keywords are available:

- | | |
|--------------------|---|
| ANALYSIS | <i>Only the results of the experimental data analysis.</i> |
| SIMULATIONS | <i>Only the results of the simulation data analysis. The results of three simulation models—maximum utility, Bradley-Terry-Luce (BTL), and logit—are displayed.</i> |

SUMMARYONLY	<i>Only the summaries in the output, not the individual subjects.</i> Thus, if you have a large number of subjects, you can see the summary results without having to generate output for each subject.
ALL	<i>The results of both the experimental data and simulation data analyses.</i> ALL is the default.
NONE	<i>No results are written to the display file.</i> This keyword is useful if you are interested only in writing the utility file (see “UTILITY Subcommand” below).

UTILITY Subcommand

UTILITY writes a utility file to the file specified. The utility file is an SPSS data file.

- If UTILITY is not specified, no utility file is written.
- UTILITY is followed by the name of the file to be written.
- The file is specified in the usual manner for your operating system.
- The utility file contains one case for each subject.

The variables written to the utility file are in the following order:

- Any SPLIT FILE variables in the working data file.
- Any SUBJECT variable.
- The constant for the regression equation for the subject. The regression equation constant is named CONSTANT.
- For DISCRETE factors, all of the utilities estimated for the subject. The names of the utilities estimated with DISCRETE factors are formed by appending a digit after the factor name. The first utility gets a 1, the second a 2, and so on.
- For LINEAR factors, a single coefficient. The name of the coefficient for LINEAR factors is formed by appending *_L* to the factor name. (To calculate the predicted score, multiply the factor value by the coefficient.)
- For IDEAL or ANTIIDEAL factors, two coefficients. The name of the two coefficients for IDEAL or ANTIIDEAL factors are formed by appending *_L* and *_Q*, respectively, to the factor name. (To use these coefficients in calculating the predicted score, multiply the factor value by the first and add that to the product of the second coefficient and the square of the factor value.)
- The estimated ranks or scores for all profiles in the plan file. The names of the estimated ranks or scores are of the form *SCORE_n* for experimental and holdout profiles, or *SIMUL_n* for simulation profiles, where *n* is the position in the plan file. The name is *SCORE* for experimental and holdout profiles even if the data are ranks.

If the variable names created are too long, letters are truncated from the end of the original variable name before new suffixes are appended.

PLOT Subcommand

The PLOT subcommand produces high-resolution plots in addition to the output usually produced by CONJOINT.

- If high-resolution graphics is turned off, the plots are not produced and a warning is displayed (see the HIGHRES subcommand of the SET command in the *SPSS Base Syntax Reference Guide*).

The following keywords are available for this subcommand:

SUMMARY	<i>Plots a high-resolution bar chart of the importance values for all variables, plus a utility bar chart for each variable. This is the default if the PLOT subcommand is specified with no keywords.</i>
SUBJECT	<i>Plots a clustered bar chart of the importance values for each factor, clustered by subjects, and one clustered bar chart for each factor showing the utilities for each factor level, clustered by subjects. If no SUBJECT subcommand was specified naming the variables, no plots are produced and a warning is displayed.</i>
ALL	<i>Plots both summary and subject charts.</i>
NONE	<i>Does not plot any high-resolution charts. This is the default if the subcommand is omitted.</i>

ORTHOPLAN

```
ORTHOPLAN [FACTORS=varlist ['labels'] (values ['labels'])...]
[{/REPLACE          }]
[{/OUTFILE=file}]
[/MINIMUM=value]
[/HOLDOUT=value]  [ /MIXHOLD={YES}]
                  {NO }
```

Example:

```
ORTHOPLAN FACTORS=SPEED 'Highest possible speed'
          (70 '70 mph' 100 '100 mph' 130 '130mph')
WARRANTY 'Length of warranty'
          ('1 year' '3 year' '5 year')
SEATS (2, 4)
/MINIMUM=9 /HOLDOUT=6.
```

Overview

ORTHOPLAN generates an orthogonal main-effects plan for a full-concept conjoint analysis. It can append or replace an existing working data file, or build a working data file if one does not already exist. The generated plan can be listed in full-concept profile, or card, format using PLANCARDS. The file created by ORTHOPLAN can be used as the plan file for CONJOINT.

Options

Number of Cases. You can specify the minimum number of cases to be generated in the plan.

Holdout and Simulation Cases. In addition to the experimental main-effects cases, you can generate a specified number of holdout cases and identify input data as simulation cases.

Basic Specification

- The basic specification is ORTHOPLAN followed by FACTORS, a variable list, and a value list in parentheses. ORTHOPLAN will generate cases in the working data file, with each case representing a profile in the conjoint experimental plan and consisting of a new combination of the factor values. By default, the smallest possible orthogonal plan is generated.
- If you are appending to an existing working data file that has previously defined values, the FACTORS subcommand is optional.

Subcommand Order

- Subcommands can be named in any order.

Operations

- ORTHOPLAN builds a working data file if one does not already exist by using the variable and value information on the FACTORS subcommand.
- When ORTHOPLAN appends to a working data file and FACTORS is not used, the factor levels (values) must be defined on a previous ORTHOPLAN or VALUE LABELS command.
- New variables *STATUS_* and *CARD_* are created and added to the working data file by ORTHOPLAN if they do not already exist. *STATUS_*=0 for experimental cases, 1 for holdout cases, and 2 for simulation cases. Holdout cases are judged by the subjects but are not used when CONJOINT estimates utilities. Instead, they are used as a check on the validity of the estimated utilities. Simulation cases are entered by the user. They are factor-level combinations that are not rated by the subjects but are estimated by CONJOINT based on the ratings of the experimental cases. *CARD_* contains the case identification numbers in the generated plan.
- Duplication between experimental and simulation cases is reported.
- If a user-entered experimental case (*STATUS_*=0) is duplicated by ORTHOPLAN, only one copy of the case is kept.
- Occasionally, ORTHOPLAN may generate duplicate experimental cases. One way to handle these duplicates is simply to edit or delete them, in which case the plan is no longer orthogonal. Alternatively, you can try running ORTHOPLAN again. With a different *seed*, ORTHOPLAN might produce a plan without duplicates. See the SEED subcommand on SET in the *SPSS Base Syntax Reference Guide* for more information on the random seed generator.
- The SPLIT FILE and WEIGHT commands are ignored by ORTHOPLAN.

Limitations

- Missing data are not allowed.
- A maximum of 10 factors and 9 levels can be specified per factor.
- A maximum of 81 cases can be generated by ORTHOPLAN.

Example

```
ORTHOPLAN FACTORS=SPEED 'Highest possible speed'
  (70 '70 mph' 100 '100 mph' 130 '130mph')
WARRANTY 'Length of warranty'
  ('1 year' '3 year' '5 year')
SEATS (2, 4) /MINIMUM=9 /HOLDOUT=6 /OUTFILE='CARPLAN.SAV' .
```

- The FACTORS subcommand defines the factors and levels to be used in building the file. Labels for some of the factors and some of the levels of each factor are also supplied.
- The MINIMUM subcommand specifies that the orthogonal plan should contain at least nine full-concept cases.
- HOLDOUT specifies that six holdout cases should be generated. A new variable, *STATUS_*, is created by ORTHOPLAN to distinguish these holdout cases from the regular experimental cases. Another variable, *CARD_*, is created to assign identification numbers to the plan cases.
- The OUTFILE subcommand saves the plan generated by ORTHOPLAN as a data file so it can be used at a later date with CONJOINT.

Example

```
DATA LIST FREE /SPEED WARRANTY SEATS.
VALUE LABELS speed 70 '70 mph' 100 '100 mph' 130 '130 mph'
/WARRANTY 1 '1 year' 3 '3 year' 5 '5 year'
/SEATS 2 '2 seats' 4 '4 seats'.
BEGIN DATA
130 5 2
130 1 4
END DATA.
ORTHOPLAN
/OUTFILE='CARPLAN.SAV'.
```

- In this example, ORTHOPLAN appends the plan to the working data file and uses the variables and values previously defined in the working data file as the factors and levels of the plan.
- The data between BEGIN DATA and END DATA are assumed to be simulation cases and are assigned a value of 2 on the newly created *STATUS_* variable.
- The OUTFILE subcommand saves the plan generated by ORTHOPLAN as a data file so it can be used at a later date with CONJOINT.

FACTORS Subcommand

FACTORS specifies the variables to be used as factors and the values to be used as levels in the plan.

- FACTORS is required for building a new working data file or replacing an existing one. It is optional for appending to an existing file.
- The keyword FACTORS is followed by a variable list, an optional label for each variable, a list of values for each variable, and optional value labels.
- The list of values and the value labels are enclosed in parentheses. Values can be numeric or they can be strings enclosed in apostrophes.
- The optional variable and value labels are enclosed in apostrophes.
- If the FACTORS subcommand is not used, every variable in the working data file (other than *STATUS_* and *CARD_*) is used as a factor, and level information is obtained from the value labels that are defined in the working data file. ORTHOPLAN *must* be able to find value information either from a FACTORS subcommand or from a VALUE LABELS command. (See the VALUE LABELS command in the *SPSS Base Syntax Reference Guide*.)

Example

```

ORTHOPLAN FACTORS=SPEED 'Highest possible speed'
      (70 '70 mph' 100 '100 mph' 130 '130mph')
      WARRANTY 'Length of warranty'
      (1 '1 year' 3 '3 year' 5 '5 year')
      SEATS 'Number of seats' (2 '2 seats' 4 '4 seats')
      EXCOLOR 'Exterior color'
      INCOLOR 'Interior color' ('RED' 'BLUE' 'SILVER').

```

- *SPEED*, *WARRANTY*, *SEATS*, *EXCOLOR*, and *INCOLOR* are specified as the factors. They are given the labels *Highest possible speed*, *Length of warranty*, *Number of seats*, *Exterior color*, and *Interior color*.
- Following each factor and its label are the list of values and the value labels in parentheses. Note that the values for two of the factors, *EXCOLOR* and *INCOLOR*, are the same and thus need to be specified only once after both factors are listed.

REPLACE Subcommand

REPLACE can be specified to indicate that the working data file, if present, should be replaced by the generated plan. There is no further specification after the REPLACE keyword.

- By default, the working data file is not replaced. Any new variables specified on a FACTORS subcommand plus the variables *STATUS_* and *CARD_* are appended to the working data file.
- REPLACE should be used when the current working data file has nothing to do with the plan file to be built. The working data file will be replaced with one that has variables *STATUS_*, *CARD_*, and any other variables specified on the FACTORS subcommand.
- If REPLACE is specified, the FACTORS subcommand is required.

OUTFILE Subcommand

OUTFILE saves the orthogonal design to an SPSS data file. The only specification is a name for the output file.

- By default, a new data file is not created. Any new variables specified on a FACTORS subcommand plus the variables *STATUS_* and *CARD_* are appended to the working data file.
- The output data file contains variables *STATUS_*, *CARD_*, and any other variables specified on the FACTORS subcommand.
- The file created by OUTFILE can be used by other SPSS commands, such as PLANCARDS and CONJOINT.
- If both OUTFILE and REPLACE are specified, REPLACE is ignored.

MINIMUM Subcommand

MINIMUM specifies a minimum number of cases for the plan.

- By default, the minimum number of cases necessary for the orthogonal plan is generated.

- MINIMUM is followed by a positive integer less than or equal to the total number of cases that can be formed from all possible combinations of the factor levels.
- If ORTHOPLAN cannot generate at least the number of cases requested on MINIMUM, it will generate the largest number it can that fits the specified factors and levels.

HOLDOUT Subcommand

HOLDOUT creates holdout cases in addition to the regular plan cases. Holdout cases are judged by the subjects but are not used when CONJOINT estimates utilities.

- If HOLDOUT is not specified, no holdout cases are produced.
- HOLDOUT is followed by a positive integer less than or equal to the total number of cases that can be formed from all possible combinations of factor levels.
- Holdout cases are generated from another random plan, not the main-effects experimental plan. The holdout cases will not duplicate the experimental cases or each other.
- The experimental and holdout cases will be randomly mixed in the generated plan or the holdout cases will be listed after the experimental cases, depending on subcommand MIXHOLD. The value of *STATUS_* for holdout cases is 1. Any simulation cases will follow the experimental and holdout cases.

MIXHOLD Subcommand

MIXHOLD indicates whether holdout cases should be randomly mixed with the experimental cases or should appear separately after the experimental plan in the file.

- If MIXHOLD is not specified, the default is NO, meaning holdout cases will appear after the experimental cases in the file.
- MIXHOLD followed by keyword YES requests that the holdout cases be randomly mixed with the experimental cases.
- MIXHOLD specified without a HOLDOUT subcommand has no effect.

PLANCARDS

```
PLANCARDS [FACTORS=varlist]

[/FORMAT={LIST}
          {CARD}
          {BOTH}]

[/TITLE='string']

[/FOOTER='string']

[/OUTFILE=file]

[/PAGINATE]
```

Example:

```
PLANCARDS FORMAT=BOTH/ OUTFILE='DESIGN.FRM'
/TITLE='Car for Sale' /FOOTER='Type )card'
/PAGINATE.
```

Overview

PLANCARDS produces full-concept profiles, or cards, from a plan file for a conjoint analysis study. The plan file can be generated by ORTHOPLAN or entered by the user. The printed profiles can be used as the experimental stimuli that subjects judge in terms of preference.

Options

Format. You can produce profiles in the usual listing-file format, in single-profile format, or both.

Titles and Footers. You can specify title and footer labels that appear at the top and bottom of the listing or, for single-card format, at the top and bottom of each profile. You can include an identifying profile number as part of the title or footer.

Pagination. You can control whether profiles written in single-profile format should begin a new page at the beginning of each profile.

Basic Specification

- The basic specification is PLANCARDS. This produces a standard listing of profiles in your listing file using all variables in the working data file except *STATUS_* and *CARD_* as factors.

Subcommand Order

- Subcommands can be named in any order.

Operations

- PLANCARDS assumes that the working data file represents a plan for a full-concept conjoint study. Each “case” in such a file is one profile in the conjoint experimental plan.
- Factor and factor-level labels in the working data file, generated by ORTHOPLAN or by the VARIABLE and VALUE LABELS commands, are used in the output.
- The SPSS command SPLIT FILE is ignored for single-profile format. In listing-file format, each subfile represents a different plan, and a new listing begins for each one.
- The WEIGHT command is ignored by PLANCARDS.

Limitations

- Missing values are not recognized as missing and are treated like other values.

Example

```

ORTHOPLAN FACTORS=SPEED 'Highest possible speed'
      (70 '70 mph' 100 '100 mph' 130 '130mph)
      WARRANTY 'Length of warranty' ('1 year' '3 year' '5 year')
      SEATS 'Number of seats' (2, 4) /MINIMUM=9 /HOLDOUT=6.
PLANCARDS FORMAT=BOTH /OUTFILE='DESIGN.FRM'
      /TITLE='Car for Sale' /FOOTER='Type )card' /PAGINATE.

```

- ORTHOPLAN generates a set of profiles (cases) for a full-concept conjoint analysis in the working data file.
- PLANCARDS produces a standard listing file containing the profiles in the output file *DESIGN.FRM*.
- Each profile in *DESIGN.FRM* will have the title *Car for Sale* at the top and the label *Type n* at the bottom, where *n* is a profile identification number.
- The PAGINATE subcommand specifies that each new profile in the *DESIGN.FRM* file should begin on a new page. This makes the profiles in the file convenient to use as the actual profiles the experimenter hands to the subjects.

Example

```

DATA LIST FREE/ COST NEWNESS EXPER NAME REP
              GUARAN TRIAL TRUST.
VARIABLE LABELS
  COST 'Product cost'
  NEWNESS 'Product newness'
  EXPER 'Brand experience'
  NAME "Manufacturer's Name"
  REP "Distributor's reputation"
  GUARAN 'Money-back Guarantee'
  TRIAL 'Free sample/trial'
  TRUST 'Endorsed by a trusted person'.
VALUE LABELS
  COST 1 'LOW' 2 'HIGH'/
  NEWNESS 1 'NEW' 2 'OLD'/
  EXPER 1 'SOME' 2 'NONE'/
  NAME 1 'ESTABLISHED' 2 'UNKNOWN'/
  REP 1 'GOOD' 2 'UNKNOWN'/
  GUARAN 1 'YES' 2 'NO'/
  TRIAL 1 'YES' 2 'NO'/
  TRUST 1 'YES' 2 'NO'.
BEGIN DATA
  1 2 2 1 2 2 2 1
  2 2 2 1 1 1 2 1
  2 2 1 2 2 1 1 1
  2 1 2 1 2 2 1 2
  2 1 1 2 2 2 2 1
  2 1 2 2 1 1 2 2
  1 1 2 2 1 2 1 1
  1 1 1 1 2 1 2 2
  1 2 1 2 1 2 2 2
  1 1 1 1 1 1 1 1
  2 2 1 1 1 2 1 2
  1 2 2 2 2 1 1 2
END DATA.
PLANCARDS TITLE=' ' 'Profile #)CARD' /FOOTER='RANK:' ' '.

```

- In this example, the plan is entered and defined by the user rather than by ORTHOPLAN.
- PLANCARDS uses the information in the working data file to produce a set of profiles in the standard listing file. See Figure 1 on p. 20 for the output produced by this command. (The variables and values in this example were taken from Akaah & Korgaonkar, 1988).

FACTORS Subcommand

FACTORS identifies the variables to be used as factors and the order in which their labels are to appear in the output. String variables are permitted.

- Keyword FACTORS is followed by a variable list.
- By default, if FACTORS is not specified, all variables in the working data file except those named *STATUS_* or *CARD_* are used as factors in the order in which they appear in the file. (See the ORTHOPLAN command for information on variables *STATUS_* and *CARD_*.)

FORMAT Subcommand

FORMAT specifies whether the profiles should use standard listing-file format, single-profile format, or both.

- The keyword FORMAT is followed by LIST, CARD, or BOTH. (ALL is an alias for BOTH.)
- The default output is LIST (listing-file format).
- With LIST format, holdout profiles are differentiated from experimental profiles, and simulation profiles are listed separately following the experimental and holdout profiles. With CARD format, holdout profiles are not differentiated and simulation profiles are not produced.
- If CARD or BOTH is specified without an OUTFILE subcommand, the single profiles are included in the listing file.

Example

```
PLANCARDS FORMAT=BOTH
  /TITLE=' ' 'Profile #)CARD' /FOOTER='RANK:'.
```

- The listing-file and single-profile output for the first two profiles are shown in Figure 1 and Figure 2.

Figure 1 Listing-file format

```
Plancards:
Title:
  Profile #)CARD
Card 1
  Product cost LOW
  Product newness OLD
  Brand experience NONE
  Manufacturer's Name ESTABLISHED
  Distributor's reputation UNKNOWN
  Money-back Guarantee NO
  Free sample/trial NO
  Endorsed by a trusted person YES
Card 2
  Product cost HIGH
  Product newness OLD
  Brand experience NONE
  Manufacturer's Name ESTABLISHED
  Distributor's reputation GOOD
  Money-back Guarantee YES
  Free sample/trial NO
  Endorsed by a trusted person YES
. . .
Footer: RANK:
```

Figure 2 Single-profile format

```

Profile #1

Product cost LOW
Product newness OLD
Brand experience NONE
Manufacturer's Name ESTABLISHED
Distributor's reputation UNKNOWN
Money-back Guarantee NO
Free sample/trial NO
Endorsed by a trusted person YES

RANK:

Profile #2

Product cost HIGH
Product newness OLD
Brand experience NONE
Manufacturer's Name ESTABLISHED
Distributor's reputation GOOD
Money-back Guarantee YES
Free sample/trial NO
Endorsed by a trusted person YES

RANK:

. . .

```

OUTFILE Subcommand

OUTFILE names an external file where profiles in single-profile format are to be written.

- By default, profiles are written to the listing file; no external file is written.
- The OUTFILE keyword is followed by the name of an external file. The file is specified in the usual manner for your system.
- Profiles are written to an external file in single-profile format unless otherwise specified on the FORMAT subcommand.

TITLE Subcommand

TITLE specifies a string to be used at the top of the output in listing format or at the top of each new profile in profile format.

- If TITLE is not used, no title appears above the first attribute.
- The keyword TITLE is followed by a string enclosed in apostrophes.
- Quotation marks can be used to enclose the string instead of apostrophes when you want to use an apostrophe in the title.
- Multiple strings per TITLE subcommand can be specified; each one will appear on a separate line.
- Use an empty string (' ') to cause a blank line.
- Multiple TITLE subcommands can be specified; each one will appear on a separate line.

- If the special character sequence)CARD is specified anywhere in the title, PLANCARDS will replace it with the sequential profile number in single-profile-formatted output. Having the profile number automatically printed on the profile will help the experimenter to record the data accurately. This character sequence is not translated in listing-file format.

FOOTER Subcommand

FOOTER specifies a string to be used at the bottom of the output in listing format or at the bottom of each profile in profile format.

- If FOOTER is not used, nothing appears after the last attribute.
- FOOTER is followed by a string enclosed in apostrophes.
- Quotation marks can be used to enclose the string instead of apostrophes when you want to use an apostrophe in the footer.
- Multiple strings per FOOTER subcommand can be specified; each one will appear on a separate line.
- Use an empty string (' ') to cause a blank line.
- Multiple FOOTER subcommands can be specified; each one will appear on a separate line.
- If the special character sequence)CARD is specified anywhere in the footer, PLANCARDS will replace it with the sequential profile number in single-profile-formatted output. Having the profile number automatically printed on the profile will help the experimenter to record the data accurately. This character sequence is not translated in listing-file format.

Example

```
PLANCARDS
TITLE='Profile # )CARD' ' '
  'Circle the number in the scale at the bottom that
  'indicates how likely you are to purchase this item.' ' '
/FOOTER= '0 1 2 3 4 5 6 7 8 9 10'
          'Not at all      May or may      Certainly'
          'likely to      not          would'
          'purchase      purchase     purchase'
          '-----'
/FORMAT=CARD.
```

The above example would produce the following output for the first profile:

Profile # 1

Circle the number in the scale at the bottom that indicates how likely you are to purchase this item.

Product cost LOW
 Product newness OLD
 Brand experience NONE
 Manufacturer's Name ESTABLISHED
 Distributor's reputation UNKNOWN
 Money-back Guarantee NO
 Free sample/trial NO
 Endorsed by a trusted person YES

0	1	2	3	4	5	6	7	8	9	10
Not at all				May or may					Certainly	
likely to				not					would	
purchase				purchase					purchase	

PAGINATE Subcommand

PAGINATE indicates that each new profile in single-profile format should begin on a new page.

- PAGINATE is ignored in listing-file format.
- If PAGINATE is not specified with the profile format, the profiles will not have carriage control characters that cause page breaks after each profile.
- PAGINATE has no additional specifications.

Subject Index

data files
 appending orthogonal designs, 15
 saving profiles in Display Design procedure, 21
Display Design procedure
 saving profiles in data files, 21
duplicate cases
 in Generate Orthogonal Design procedure, 13

Generate Orthogonal Design procedure
 appending to working data files, 15
 duplicate cases, 13

profiles
 saving in data files, 21

working data files
 appending orthogonal designs, 15

Syntax Index

- A
- ALL (keyword)
 - CONJOINT command 10
- ANALYSIS (keyword)
 - CONJOINT command 9
- ANTIIDEAL (keyword)
 - CONJOINT command 8
- B
- BOTH (keyword)
 - PLANCARDS command 20
- C
-)CARD (keyword)
 - PLANCARDS command 21
- CARD (keyword)
 - PLANCARDS command 20
- CONJOINT (command) 1
 - DATA subcommand 5
 - FACTOR subcommand 7
 - PLAN subcommand 4
 - PRINT subcommand 9
 - RANK subcommand 6
 - SCORE subcommand 6
 - SEQUENCE subcommand 6
 - SUBJECT subcommand 7
 - UTILITY subcommand 10
 - with ORTHOPLAN command 1, 4
- D
- DATA (subcommand)
 - CONJOINT command 5
 - with PLAN subcommand 5
- DISCRETE (keyword)
 - CONJOINT command 8
- F
- FACTOR (subcommand)
 - CONJOINT command 7
 - with UTILITY subcommand 10
- FACTORS (subcommand)
 - ORTHOPLAN command 14
 - PLANCARDS command 19
 - with REPLACE subcommand 15
- FOOTER (subcommand)
 - PLANCARDS command 22
- FORMAT (subcommand)
 - PLANCARDS command 20
- H
- HOLDOUT (subcommand)
 - ORTHOPLAN command 16
 - with MIXHOLD subcommand 16
- I
- IDEAL (keyword)
 - CONJOINT command 8
- L
- LESS (keyword)
 - CONJOINT command 8
- LINEAR (keyword)
 - CONJOINT command 8
- LIST (keyword)
 - PLANCARDS command 20
- M
- MINIMUM (subcommand)
 - ORTHOPLAN command 15
- MIXHOLD (subcommand)
 - ORTHOPLAN command 16
 - with HOLDOUT subcommand 16
- MORE (keyword)
 - CONJOINT command 8
- N
- NONE (keyword)
 - CONJOINT command 10
- O
- ORTHOPLAN (command) 12
 - CARD_ variable 13
 - duplicate cases 13
 - FACTORS subcommand 14
 - holdout cases 13
 - HOLDOUT subcommand 16
 - minimum number of cases 15
 - MINIMUM subcommand 15
 - MIXHOLD subcommand 16
 - REPLACE subcommand 15

- replacing active system file 15
- STATUS_ variable 13
- value labels 14
- with CONJOINT command 1, 4
- with PLANCARDS command 18
- with SET SEED command 13
- with VALUE LABELS command 14

OUTFILE (subcommand)
 PLANCARDS command 21

P

PAGINATE (subcommand)
 PLANCARDS command 23

PLAN (subcommand)
 CONJOINT command 4
 with DATA subcommand 5

PLANCARDS (command) 17
 FACTORS subcommand 19
 FOOTER subcommand 22
 FORMAT subcommand 20
 OUTFILE subcommand 21
 PAGINATE subcommand 23
 sequential profile numbers 22
 TITLE subcommand 21
 with ORTHOPLAN command 18
 with VALUE LABELS command 18
 with VARIABLE LABELS command 18

PRINT (subcommand)
 CONJOINT command 9

R

REPLACE (subcommand)
 ORTHOPLAN command 15
 with FACTORS subcommand 15

S

SEQUENCE (subcommand)
 CONJOINT command 6

SIMULATIONS (keyword)
 CONJOINT command 9

SUBJECT (subcommand)
 CONJOINT command 7

T

TITLE (subcommand)
 PLANCARDS command 21

U

UTILITY (subcommand)
 CONJOINT command 10
 with FACTOR subcommand 10

V

VALUE LABELS (command)
 with ORTHOPLAN command 14
 with PLANCARDS command 18

VARIABLE LABELS (command)
 with PLANCARDS command 18